

Programování 6.L, 2.A

Úkol od 7. 5. do 13. 5.

Quicksort

Metodu třídění (seřazování) zvanou Quicksort po vás nebudu chtít vymyslet, ale vzorově ji uvedu:

```
program Rychle_trideni;
uses Crt;
type pole=array [1..10] of integer;
var n:integer;
    a:pole;

procedure quicksort(var s:pole;d,h:integer);
var i,j,x,y:integer;
begin
    i:=d;
    j:=h;
    x:=s[(d+h) div 2];
    repeat
        while s[i]<x do i:=i+1;
        while s[j]>x do j:=j-1;
        if i<=j then begin
            y:=s[i];
            s[i]:=s[j];
            s[j]:=y;
            i:=i+1;
            j:=j-1;
        end;
    until i>j;
    if d<j then quicksort(s,d,j);
    if i<h then quicksort(s,i,h);
end;

begin
    clrscr;
    writeln('==== Q U I C K S O R T ===');
    writeln('Zadej 10 cisel');
    for n:=1 to 10 do begin
        write(n, '. cislo: ');
        readln(a[n]);
    end;
    quicksort(a,1,10);
    writeln('Serazena cisla: ');
    for n:=1 to 9 do write(a[n], ', ');
    writeln(a[10]);
    readln;
end.
```

Úkoly

- 1) Prostudujte si program a hlavně proceduru quicksort – kousek po kousku. Snažte se pochopit všechny řádky. Všimněte si typů parametrů a dvojího použití rekurze na konci procedury.
- 2) Na papír (!) si napište pod sebe $s[1] =$, $s[2] =$, $s[3] =$, $s[4] =$, $s[5] =$, $s[6] =$, $s[7] =$, $s[8] =$, $s[9] =$, $s[10] =$, $d =$, $h =$, $i =$, $j =$, $x =$ (y nemusíte, je to pouze pomocná proměnná pro přehození hodnot $s[i]$ a $s[j]$).
- 3) Vstupte do procedury a napište k proměnným prvotní hodnoty – $s[1]$ až $s[10]$ náhodná neseřazená celá čísla, $d = 1$, $h = 10$.
- 4) Procházejte procedurou poctivě krok za krokem a na papíře měňte hodnoty proměnných (vedle staré hodnoty napište novou hodnotu, starou škrtněte). Správně se rozhodujte u všech podmínek, nepřeskakujte nic, ani cyklus, i kdyby to trvalo dlouho (nebude, je to quicksort ☺), jinak by to ztratilo smysl.
- 5) Pozor je třeba dát na rekurzi – zatímco pole s jako parametr volaný odkazem zůstává při všech voláních rekurze totéž, parametry volané hodnotou d , h a lokální proměnné i , j budou pokaždé nové! Na papíře si vedle nich udělejte svislou čáru a za ní začněte s novými hodnotami. Po ukončení procedury tyhle hodnoty škrtněte a vraťte se k původním hodnotám před čárou.
- 6) Pokud se vám celý postup povede bez chyby, budete mít na konci seřazené hodnoty v poli s .
- 7) Teprve potom program přepište/zkopírujte do TP a vyzkoušejte jeho funkčnost.

Celý svůj ofocený papírový postup **pošlete** na an@glp.cz nejpozději **ve středu 13. 5.**

Další úkoly zveřejním na webu školy ve čtvrtek 14. 5.

Filip Andziol